1.) Polynome

Aufgaben: 3

```
> restart;
```

Definition

Ein zentrales Element in vielen Bereichen der Mathematik sind Polynome, z.B. verhalten sich sie in der Analysis meistens gutartig, sprich liefern durch induzierte Abbildungen eine ganze Beispielklasse für Eigenschaften wie Stetigkeit und Differenzierbarkeit, welche du demnächst kennenlernen wirst. In der linearen Algebra sind einige Objekte von Interesse durch Polynome gegeben, weshalb wir an dieser Stelle Polynome formalisieren und etwas genauer betrachten wollen.

MATH: Sei K ein kommutativer Ring mit 1. Wir betrachten K-wertige Folgen

$$a:\mathbb{Z}_{\geq 0} \to K: i \mapsto a_i,$$

so dass ein von a abhängiges $n \in \mathbb{N}$ existiert mit $a_i = 0$ für alle $i > n$. Das kleinste derartige n heißt auch der **Grad** von a. Nur der Nullfolge

$$0:\mathbb{Z}_{\geq 0} \to K: i \mapsto 0$$

ordnet man keinen Grad zu. Jede derartige Folge nennen wir ein **Polynom** über K.

Klar: Da K ein Ring ist, kann man Polynome werteweise addieren:

$$a:\mathbb{Z}_{\geq 0} \to K: i \mapsto a_i$$

$$b:\mathbb{Z}_{\geq 0} \to K: i \mapsto b_i$$

Dann ist ihre Summe so definiert:

$$a+b:\mathbb{Z}_{\geq 0} \to K: i \mapsto a_i + b_i.$$

Mit dieser Addition haben wir eine abelsche Gruppe.

Wir wollen auch eine Multiplikation einführen und zwar so, dass

$$(1, 0, 0, 0, \ldots)$$

das Einselement ist und die Multiplikation mit

$$(0, 1, 0, 0, \ldots)$$

für

$$a = (a_0, a_1, a_2, \ldots)$$

eine Verschiebung nach rechts ergibt:

$$(0, 1, 0, 0, \ldots) \cdot (a_0, a_1, a_2, \ldots) := (0, a_0, a_1, a_2, \ldots).$$

Hier ist die Formel für das Produkt:

$$a \cdot b:\mathbb{Z}_{\geq 0} \to K: i \mapsto a_0 \cdot b_i + a_1 \cdot b_{i-1} + a_2 \cdot b_{i-2} + \ldots + a_i \cdot b_0.$$

KOMMENTAR: Falls dich dies an das Cauchy-Produkt erinnert, so liegst du richtig. Durch die "Endlichkeit" der Folge sind alle Reihen zu Polynomen als endliche Summen absolut konvergent.

NOTATION: Sowohl in der Mathematik als auch in Maple ist es unüblich, Polynome als Folgen zu schreiben, obschon sie es sind. Es ist praktischer, das Polynom $(a_0, a_1, \ldots, a_n, 0, 0, \ldots)$ als

```
> Sum(a[i]*x^i,i=0..n);
```

$$\sum_{i=0}^{n} a_i x^i \tag{1.1.1}$$

zu schreiben. (**DENKANSTOSS**: Identifiziere das Polynom x mit der zugehörigen Folge. Als welche Folgen kann man Elemente des Ringes/Körpers auffassen? Erkenne den Zusammenhang zwischen diesen besonderen Folgen und der Summenschreibweise.)

```
> A := sum(a[i]*x^i,i=0..10);
```

$$A := x^{10} a_{10} + x^9 a_9 + x^8 a_8 + x^7 a_7 + x^6 a_6 + x^5 a_5 + x^4 a_4 + x^3 a_3 + x^2 a_2 + x a_1 + a_0 \tag{1.1.2}$$

```
> B := sum(b[i]*x^i,i=0..4);
```

$$B := x^4 b_4 + x^3 b_3 + x^2 b_2 + x b_1 + b_0 \tag{1.1.3}$$

```
> for j from 0 to 5 do
    coeff(A*B,x,j);
  end do;
  j:='j':
```

$$a_0 b_0$$

$$a_0 b_1 + a_1 b_0$$

$$a_0 b_2 + a_1 b_1 + a_2 b_0$$

$$a_0 b_3 + a_1 b_2 + a_2 b_1 + a_3 b_0$$

$$a_0 b_4 + a_1 b_3 + a_2 b_2 + a_3 b_1 + a_4 b_0$$

$$a_1 b_4 + a_2 b_3 + a_3 b_2 + a_4 b_1 + a_5 b_0 \tag{1.1.4}$$

MAPLE: Der Befehl **coeff(f,x,j)** gibt das j-te Folgenglied aus der Darstellung von f als Summe von Potenzen von x zurück.

MATH: Man kann nun nachrechnen, dass die Menge aller Polynome über K einen kommutativen Ring mit Eins bildet. Dieser heißt **Polynomring** über K und wird mit $K[x]$ bezeichnet.

Einsetzungshomomorphismus und Polynomfunktionen

MATH: Ist $K[x]$ ein Polynomring über dem Ring K, so bekommt man für jedes $r \in K$ einen Ringhomomorphismus (genannt **Einsetzungshomomorphismus**):

$$\eta_r : K[x] \to K : \sum_{i=0}^{n} a_i \cdot x^i \mapsto \sum_{i=0}^{n} a_i \cdot r^i$$

MATH: Bezeichnet man die linke Seite mit $p = p(x)$, so bezeichnet man häufig die rechte Seite mit $p(r)$. D.h. man hat für die Unbestimmte x die konkrete Zahl r eingesetzt. Die Ringhomomorphieeigenschaften lesen sich dann so:

$(p+q)(r) = p(r) + q(r)$
$(p \cdot q)(r) = p(r) \cdot q(r)$
$1(r) = 1$

MATH: Sei nun K ein Körper. Man kann Polynome aus $K[x]$ dazu benutzen, um gewisse Abbildungen von K nach K darzustellen. Z.B. stellt das Polynom x immer die Identitätsabbildung von K nach K dar.
Allgemeiner stellt $p(x) \in K[x]$ die Abbildung

$\underline{p}: K \to K : r \mapsto p(r)$

dar, genannt die durch p **induzierte Polynomabbildung**.

MATH: $K[x] \to K^K : p \mapsto \underline{p}$ ist ein Ringhomomorphismus. Sein Bild in K^K heißt der **Ring der Polynomfunktionen**.

DENKANSTOSS: Beweise dies.

BEISPIEL: $K := \mathbb{Z}/2\mathbb{Z}$. Die Abbildung $K \to K$, die die beiden Elemente vertauscht, wird durch $x+1$ dargestellt.

ÜBUNG [01]:

1.) Finde ein Polynom vom Grad 2, welches die Nullabbildung $K \to K : k \mapsto 0$ von $K = \mathbb{Z}/2\mathbb{Z}$ darstellt.
2.) Finde ein Polynom vom Grad 3, welches die Nullabbildung $K \to K : k \mapsto 0$ von $K = \mathbb{Z}/3\mathbb{Z}$ darstellt.
3.) Sei K ein endlicher Körper. Finde ein Polynom vom Grad $|K|$, welches die Nullabbildung $K \to K : k \mapsto 0$ darstellt.
Hinweis: Worksheet 6, Übung 5, Teil 2.
4.) Zeige: Für $K = \mathbb{Z}/2\mathbb{Z}$ und $K = \mathbb{Z}/3\mathbb{Z}$ ist der Homomorphismus $K[x] \to K^K : p \mapsto \underline{p}$ ein Ringepimorphismus. (freiwillig: Für jeden endlichen Körper K.)
5.) (freiwillig) Zeige: Für jeden unendlichen Körper K ist $K[x] \to K^K : p \mapsto \underline{p}$ ein Ringmonomorphismus.
6.) Gib einen Körper K und eine Abbildung $f: K \to K$ an, die keine Polynomfunktion ist.
Hinweis: Lies drei weitere Zeilen im Worksheet.

MATH: Bei Polynomen aus $K[x]$ liegt eine Nullstelle $a \in K$ genau dann vor, wenn $(x-a)$ ein Teiler von $p(x)$ ist: $p(a) = 0 \Leftrightarrow (x-a) | p(x)$. Dies sieht man z. B. an dem Isomorphismus $K[x] \to K[x] : \sum_{i=0}^{n} b_i \cdot x^i \mapsto \sum_{i=0}^{n} b_i \cdot (x-a)^i$.

```
> p:=x^3+x^2-2;
```

$$p := x^3 + x^2 - 2 \qquad (1.2.1)$$

```
> subs(x=1,p);
```

$$0 \tag{1.2.2}$$

```
> subs(x=``(x-1),expand(subs(x=x+1,p)));
  expand( % ) = p;
```

$$(x-1)^3 + 4\,(x-1)^2 + 5\,(x-1)$$

$$x^3 + x^2 - 2 = x^3 + x^2 - 2 \tag{1.2.3}$$

MATH: Wir folgern daraus und mithilfe der Addition der Grade bei der Multiplikation der Polynome: Ein Polynom in einer Variablen hat höchstens so viele Nullstellen wie sein Grad. Insbesondere ist ein Polynom in einer Variablen mit unendlich vielen Nullstellen schon das Nullpolynom.

DENKANSTOSS: Wie viele Nullstellen hat das Polynom $2 \cdot x \in \mathbb{Z}/4\,\mathbb{Z}[x]$?

MATH: Auch Polynome in $K[x, y]$ können Nullstellen haben, die dann in $K \times K = K^2$ liegen, denn wir haben den Einsetzungshomomorphismus für jedes $(a, b) \in K^2$:

$K[x, y] \rightarrow K : p(x, y) \mapsto p(a, b)$

```
> p:=(x+y)^3-(x-y+1)^4;
  q:=(x-3*y-1)^5+3;
```

$$p := (x+y)^3 - (x-y+1)^4$$

$$q := (x - 3\,y - 1)^5 + 3 \tag{1.2.4}$$

```
> subs([x=1,y=2],p*q);
```

$$-209871 \tag{1.2.5}$$

```
> subs([x=1,y=2],p)*subs([x=1,y=2],q);
```

$$-209871 \tag{1.2.6}$$

MATH: Leider kann man bei einer Nullstelle (a, b) eines Polynoms in zwei Variablen keinen Faktor abspalten. Z. B. hat $x + y$ unendlich viele Nullstellen (falls der Körper unendlich viele Elemente besitzt). Es hat aber Grad 1 und kann daher nicht mehr faktorisiert werden. Insbesondere kann man aus der Tatsache, dass ein Polynom unendlich viele Nullstellen hat nicht schließen, dass es Null ist. Eine Analogie zu Polynomen in einer Variablen ist die folgende:

MATH - Satz: Sei $p(x, y) \in K[x, y]$ mit $p(a, b) = 0$ für alle $(a, b) \in A \times B$ mit $A, B \subseteq K$ unendlich. Dann gilt $p(x, y) = 0$.

MATH - Bemerkung: Diese Aussage wird oft in der folgenden Form benutzt:

$p(a, b) = q(a, b)\ \forall\ (a, b) \in A \times B$ wie oben $\Rightarrow$ $p = q$.

Beweis: Schreibe

$p(x, y) = \sum_{i=0}^{n} p_i(x) \cdot y^i$, also fasse das Polynom als Polynom in y auf mit Koeffizienten, welche Polynome in x sind.

Das Polynom $p(a, y) \in K[y]$ hat für jedes $a \in A$ unendlich viele Nullstellen $b \in K$, ist also gleich Null für jedes $a \in A$, d. h. $p_i(a) = 0$ für unendlich viele $a \in A$. Wegen $p_i(x) \in K[x]$, folgt $p_i(x) = 0$ für alle i, also $p(x, y) = 0$.

```
> p:=(x+y)^2-(x-y)^2;
```

$$p := (x+y)^2 - (x-y)^2 \quad (1.2.7)$$

```
> q:=4*x*y;
```

$$q := 4\,x\,y \quad (1.2.8)$$

```
> map(a->map(b->subs([x=a,y=b],p),[-1,0,1]),[-1,0,1]);
```

$$[[4, 0, -4], [0, 0, 0], [-4, 0, 4]] \quad (1.2.9)$$

```
> map(a->map(b->subs([x=a,y=b],q),[-1,0,1]),[-1,0,1]);
```

$$[[4, 0, -4], [0, 0, 0], [-4, 0, 4]] \quad (1.2.10)$$

Da beide Polynome den Grad 2 haben, zeigt diese Auswertung bereits: $p = q$.

ÜBUNG [02]:
Formuliere und beweise eine Verschärfung des letzten Satzes, der mit endlichen Mengen A und B auskommt. (Hinweis: Benutze den Grad von *p*.)

2.) Matrizen und der Gaußsche Algorithmus

Aufgaben: 5

```
> restart;
  with(LinearAlgebra):
```

Gleichungssysteme

In der Mathematik ist man oftmals an den Nullstellen eines Polynoms oder sogar den gemeinsamen Nullstellen mehrerer Polynome, eines so genannten Gleichungssystems, interessiert. Während die allgemeine Frage schwierig zu lösen ist, so gibt es doch einen einfachen Spezialfall:

MATH: Ein **lineares Gleichungssystem** über dem Körper K in n Unbekannten ist gegeben durch Polynome $f_1,..., f_m \in K[x_1,..., x_n]$, wobei für alle $1 \le i \le m$ gilt:

$f_i = c_i + \sum_{j=1}^{n} a_{i,j} \cdot x_j$, wobei für die Koeffizienten c_i, $a_{i,j} \in K$ gilt. Solche Polynome heißen auch **linear**. Eine Lösung des linearen Gleichungssystems ist ein Tupel $(a_1,..., a_n) \in K^n$ mit der Eigenschaft: $\forall\ 1 \le i \le m : f_i(a_1,..., a_n) = 0$.

Es hat sich eingebürgert, lineare Gleichungssysteme durch Matrizen darzustellen:

MATH: Eine $n \times m$-**Matrix** über einem Körper (oder Ring) K ist eine Abbildung

$$A:\{1,..,n\}\times\{1,..,m\}\to K,$$
wobei der Funktionswert von (i,j) meistens als $A_{i,j}$ geschrieben wird. In diesem Sinne sind Matrizen einfach Tabellen und insofern vielleicht nicht so interessant. Es sei denn, man kann etwas damit machen.
Die Menge aller $n\times m$-Matrizen bezeichnen wir mit $K^{n\times m}$.

Wir werden in diesem Abschnitt nur Matrizen über $K=\mathbb{R}$ betrachten. Die hier behandelten Eigenschaften von Matrizen gelten aber für beliebige Körper und nicht nur für die reellen Zahlen.

Wir können auch für Matrizen eine Multiplikation definieren, welche uns zuerst einmal eine einfachere Notation für lineare Gleichungssysteme liefert, wir aber später als sehr viel mehr erkennen werden.

MATH: Es seien $A\in K^{n\times m}$, $B\in K^{m\times l}$ zwei Matrizen. Die **Produktmatrix** $A.B$ von A und B ist definiert durch: $(A.B)_{i,j} := \sum_{k=1}^{m} A_{i,k}\cdot B_{k,j}$.

MATH: Ein **lineares Gleichungssystem** über dem Körper K kann immer auf die Gestalt
$$A.x=b$$
gebracht werden, wo $A\in K^{m\times n}$ und $b\in K^{m\times 1}$ vorgegeben sind und alle $x\in K^{n\times 1}$, die der Gleichung genügen, gesucht sind. Falls ein lineares Gleichungssystem durch $f_1,\ldots,f_m\in K[x_1,\ldots,x_n]$ gegeben sind, so erhält man A, b wie folgt:
$$A_{i,j} := coeff(f_i,x_j) = a_{i,j}$$
$$b_i := -f_i(0,\ldots,0) = negatives\ absolutes\ Glied\ von\ f_i = c_i.$$

DENKANSTOSS: Mache dir klar, dass das Einsetzen in die linearen Polynome dasselbe wie die Matrixmultiplikation der entsprechenden Matrix mit der Lösung(smatrix) ist.

Matrizen induzieren Abbildungen

MATH: Besondere Matrizen sind die **Spalten**, aufgefasst als Abbildungen $\{1,..,n\}$- >= statt $\{1,..,n\}\times\{1\}$- >= . Was (reelle) Matrizen erst richtig interessant macht, ist die Tatsache, dass sie Abbildungen von Spalten definieren:

```
> A:=Matrix(3,3,symbol=a);
```

$$A:=\begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3}\\ a_{2,1} & a_{2,2} & a_{2,3}\\ a_{3,1} & a_{3,2} & a_{3,3}\end{bmatrix} \tag{2.2.1}$$

MAPLE ist an dieser Stelle nicht konform mit der mathematischen Notation, die es ja nahegelegt hätte, symbol=A zu schreiben. Wir werden es vermutlich überleben.
Man kann auch mit symbolischen Spalten arbeiten. Statt **Matrix(n,1,...)** zu schreiben, kann man auch **Vector(n,...)** verwenden, um die unnötigen Indizes zu unterdrücken.

```
> S:=Vector(3,symbol=s);
```

$$S := \begin{bmatrix} s_1 \\ s_2 \\ s_3 \end{bmatrix} \tag{2.2.2}$$

```
> A.S;
```

$$\begin{bmatrix} a_{1,1}\,s_1 + a_{1,2}\,s_2 + a_{1,3}\,s_3 \\ a_{2,1}\,s_1 + a_{2,2}\,s_2 + a_{2,3}\,s_3 \\ a_{3,1}\,s_1 + a_{3,2}\,s_2 + a_{3,3}\,s_3 \end{bmatrix} \tag{2.2.3}$$

Die Matrix A induziert somit eine Abbildung gegeben durch

$$\phi_A : \mathbb{R}^{3\times 1} \rightarrow \mathbb{R}^{3\times 1} : S \mapsto A.S$$

KOMMENTAR: Du wirst diese Abbildungen als besonders wichtige Abbildungen im Zusammenhang mit Vektorräumen demnächst kennenlernen.

MATH: Bei einer neuen Konstruktion, z.B. der durch eine Matrix induzierten Abbildung, interessiert man sich zunächst immer für ihre Eigenschaften. Hervorstechendes Merkmal von $\mathbb{R}$-Spalten gleicher Länge ist, dass man sie addieren kann. Auch MAPLE weiß das:

```
> T:=Vector(3,symbol=t);
```

$$T := \begin{bmatrix} t_1 \\ t_2 \\ t_3 \end{bmatrix} \tag{2.2.4}$$

```
> S+T;
```

$$\begin{bmatrix} s_1 + t_1 \\ s_2 + t_2 \\ s_3 + t_3 \end{bmatrix} \tag{2.2.5}$$

MATH: Die Addition von Spalten ist mit dem Anwenden der induzierten Abbildung, also mit der Multiplikation mit A, wie man sagt, *verträglich* in folgendem Sinne:

$$A.(S+T) = A.S + A.T.$$

Das gilt auch in Maple:

```
> simplify(A.(S+T)-(A.S+A.T));
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \quad (2.2.6)$$

Ferner kann man Spalten auch mit reellen Zahlen multiplizieren:

```
> t*S;
```

$$\begin{bmatrix} t\,s_1 \\ t\,s_2 \\ t\,s_3 \end{bmatrix} \quad (2.2.7)$$

ÜBUNG [03]:

Zeige, dass das Anwenden der induzierten Abbildung mit der gerade demonstrierten Multiplikation von Spalten mit reellen Zahlen verträglich ist, d.h. dass

$$A.(t \cdot S) = t \cdot (A.S)$$

gilt.

Komposition von induzierten Abbildungen

MATH: Die Identitätsabbildung von $\mathbb{R}^{n \times 1}$ wird auch von einer Matrix induziert, nämlich der Einheitsmatrix:

```
> IdentityMatrix(3);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (2.3.1)$$

```
> IdentityMatrix(3).Vector(3,symbol=a) = Vector(3,symbol=a);
```

$$\begin{bmatrix} a_1 \\ a_2 \\ a_3 \end{bmatrix} = \begin{bmatrix} a_1 \\ a_2 \\ a_3 \end{bmatrix} \quad (2.3.2)$$

MATH: Die Spalten der $n \times n$-Einheitmatrix helfen uns, die von $A \in \mathbb{R}^{m \times n}$ induzierte Abbildung

$$\phi_A : \quad \mathbb{R}^{n \times 1} \to \mathbb{R}^{m \times 1} : S \mapsto A \cdot S$$

besser zu verstehen: In der i-ten Spalte von A ist gerade das Bild der i-ten Spalte der $n \times n$-Einheitmatrix unter dieser Abbildung:

DENKANSTOSS: Mache dir klar, dass $\phi : \mathbb{R}^{m \times n} \rightarrow Abb(\mathbb{R}^{n \times 1}, \mathbb{R}^{m \times 1}) : A \mapsto \phi_A$ eine injektive Abbildung von der Menge der Matrizen bestimmter Größe in die Menge der Abbildungen der entsprechend großen Spalten ist.

```
> A:=Matrix(3,2,symbol=a);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix} \tag{2.3.3}$$

```
> E:=IdentityMatrix(2,2);
```

$$E := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \tag{2.3.4}$$

```
> map(i->Column(E,i),[1,2]);
```

$$\left[\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix} \right] \tag{2.3.5}$$

```
> map(i->A.Column(E,i),[1,2]);
```

$$\left[\begin{bmatrix} a_{1,1} \\ a_{2,1} \\ a_{3,1} \end{bmatrix}, \begin{bmatrix} a_{1,2} \\ a_{2,2} \\ a_{3,2} \end{bmatrix} \right] \tag{2.3.6}$$

```
> map(i->Column(A,i),[1,2]);
```

$$\left[\begin{bmatrix} a_{1,1} \\ a_{2,1} \\ a_{3,1} \end{bmatrix}, \begin{bmatrix} a_{1,2} \\ a_{2,2} \\ a_{3,2} \end{bmatrix} \right] \tag{2.3.7}$$

MATH: Ist $A \in \mathbb{R}^{m \times n}$, also eine $m \times n$-Matrix über $\mathbb{R}$, und $B \in \mathbb{R}^{k \times m}$, so induzieren A und B Abbildungen

$\phi_A : \mathbb{R}^{n \times 1} \rightarrow \mathbb{R}^{m \times 1} : S \mapsto A.S$ bzw. $\phi_B : \mathbb{R}^{m \times 1} \rightarrow \mathbb{R}^{k \times 1} : T \mapsto B.T$,

die man nacheinander ausführen kann. Die Komposition ist wieder eine induzierte Abbbildung, und wird von der Produktmatrix $B.A$ induziert:

$\phi_B \circ \phi_A = \phi_{B.A}$

```
> A:=Matrix(3,2,symbol=a);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix} \tag{2.3.8}$$

```
> B:=Matrix(4,3,symbol=b);
```

$$B := \begin{bmatrix} b_{1,1} & b_{1,2} & b_{1,3} \\ b_{2,1} & b_{2,2} & b_{2,3} \\ b_{3,1} & b_{3,2} & b_{3,3} \\ b_{4,1} & b_{4,2} & b_{4,3} \end{bmatrix} \quad (2.3.9)$$

```
> B.A;
```

$$\begin{bmatrix} a_{1,1} b_{1,1} + a_{2,1} b_{1,2} + a_{3,1} b_{1,3} & a_{1,2} b_{1,1} + a_{2,2} b_{1,2} + a_{3,2} b_{1,3} \\ a_{1,1} b_{2,1} + a_{2,1} b_{2,2} + a_{3,1} b_{2,3} & a_{1,2} b_{2,1} + a_{2,2} b_{2,2} + a_{3,2} b_{2,3} \\ a_{1,1} b_{3,1} + a_{2,1} b_{3,2} + a_{3,1} b_{3,3} & a_{1,2} b_{3,1} + a_{2,2} b_{3,2} + a_{3,2} b_{3,3} \\ a_{1,1} b_{4,1} + a_{2,1} b_{4,2} + a_{3,1} b_{4,3} & a_{1,2} b_{4,1} + a_{2,2} b_{4,2} + a_{3,2} b_{4,3} \end{bmatrix} \quad (2.3.10)$$

Man sieht, dass die Spalten der Produktmatrix gleich dem Produkt von B mit den entsprechenden Spalten von A sind, z. B. ist die erste Spalte:

```
> Column(A,1), B.Column(A,1);
```

$$\begin{bmatrix} a_{1,1} \\ a_{2,1} \\ a_{3,1} \end{bmatrix}, \begin{bmatrix} a_{1,1} b_{1,1} + a_{2,1} b_{1,2} + a_{3,1} b_{1,3} \\ a_{1,1} b_{2,1} + a_{2,1} b_{2,2} + a_{3,1} b_{2,3} \\ a_{1,1} b_{3,1} + a_{2,1} b_{3,2} + a_{3,1} b_{3,3} \\ a_{1,1} b_{4,1} + a_{2,1} b_{4,2} + a_{3,1} b_{4,3} \end{bmatrix} \quad (2.3.11)$$

DENKANSTOSS: Oben haben wir eigentlich nur gezeigt: Wenn die Komposition $\phi_B \circ \phi_A$ durch eine Matrix induziert ist, dann muss die Matrix so aussehen wie angegeben. Führe die Überlegung weiter und zeige, dass die Matrix $B \cdot A$ wirklich die Abbildung $\phi_B \circ \phi_A$ induziert.

MATH: Eine grundlegende Eigenschaft des Matrixproduktes ist die Assoziativität, die sofort aus der Assoziativität der Komposition von Abbildungen folgt.

MATH: Eine besondere Rolle unter den Matrizen spielen Spalten(-Matrizen) und Zeilen(-Matrizen).
Die Produkte haben folgende Form:
n-Zeile . n-Spalte = Zahl
n-Spalte . m-Zeile = $n \times m$-Matrix

```
> Z:=Vector[row](10,i->i);
```

$$Z := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 \end{bmatrix} \quad (2.3.12)$$

```
> S:=Vector[column](10,i->i);
```

$$S := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \\ 7 \\ 8 \\ 9 \\ 10 \end{bmatrix} \quad \textbf{(2.3.13)}$$

```
> Z.S;
```

$$385 \quad \textbf{(2.3.14)}$$

```
> S.Z;
```

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 \\ 2 & 4 & 6 & 8 & 10 & 12 & 14 & 16 & 18 & 20 \\ 3 & 6 & 9 & 12 & 15 & 18 & 21 & 24 & 27 & 30 \\ 4 & 8 & 12 & 16 & 20 & 24 & 28 & 32 & 36 & 40 \\ 5 & 10 & 15 & 20 & 25 & 30 & 35 & 40 & 45 & 50 \\ 6 & 12 & 18 & 24 & 30 & 36 & 42 & 48 & 54 & 60 \\ 7 & 14 & 21 & 28 & 35 & 42 & 49 & 56 & 63 & 70 \\ 8 & 16 & 24 & 32 & 40 & 48 & 56 & 64 & 72 & 80 \\ 9 & 18 & 27 & 36 & 45 & 54 & 63 & 72 & 81 & 90 \\ 10 & 20 & 30 & 40 & 50 & 60 & 70 & 80 & 90 & 100 \end{bmatrix} \quad \textbf{(2.3.15)}$$

ÜBUNG [04]:

1) Zeige mit MAPLE die Assoziativität des Matrixproduktes:

$$(A.B).C = A.(B.C)$$

wo $A, B, C \in \mathbb{R}^{2 \times 2}$.

2) Warum ist der Beweis mit Hilfe der induzierten Abbildungen, der dies einfach als Konsequenz der Assoziativität der Komposition von Abbildungen liefert, dem in Teil 1) Gezeigten überlegen?

Wir schließen unsere Betrachtungen über Produkte von Matrizen mit alternativen Betrachtungsweisen:

```
> A:=Matrix(3,2,symbol=a);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix} \qquad \textbf{(2.3.16)}$$

```
> B:=Matrix(4,3,symbol=b);
```

$$B := \begin{bmatrix} b_{1,1} & b_{1,2} & b_{1,3} \\ b_{2,1} & b_{2,2} & b_{2,3} \\ b_{3,1} & b_{3,2} & b_{3,3} \\ b_{4,1} & b_{4,2} & b_{4,3} \end{bmatrix} \qquad \textbf{(2.3.17)}$$

MATH: Der Eintrag $(B.A)_{ij}$ ist gleich i-te Zeile von B mal j-te Spalte von A:

```
> Row(B,1).Column(A,2);
```

$$a_{1,2}\, b_{1,1} + a_{2,2}\, b_{1,2} + a_{3,2}\, b_{1,3} \qquad \textbf{(2.3.18)}$$

MATH: Die j-te Spalte von $B.A$ ist B mal j-te Spalte von A:

```
> B.Column(A,2);
```

$$\begin{bmatrix} a_{1,2}\, b_{1,1} + a_{2,2}\, b_{1,2} + a_{3,2}\, b_{1,3} \\ a_{1,2}\, b_{2,1} + a_{2,2}\, b_{2,2} + a_{3,2}\, b_{2,3} \\ a_{1,2}\, b_{3,1} + a_{2,2}\, b_{3,2} + a_{3,2}\, b_{3,3} \\ a_{1,2}\, b_{4,1} + a_{2,2}\, b_{4,2} + a_{3,2}\, b_{4,3} \end{bmatrix} \qquad \textbf{(2.3.19)}$$

MATH: Die i-te Zeile von $B.A$ ist i -te Zeile von B mal A:

```
> Row(B,1).A;
```

$$\begin{bmatrix} a_{1,1}\, b_{1,1} + a_{2,1}\, b_{1,2} + a_{3,1}\, b_{1,3} & a_{1,2}\, b_{1,1} + a_{2,2}\, b_{1,2} + a_{3,2}\, b_{1,3} \end{bmatrix} \qquad \textbf{(2.3.20)}$$

MATH: $B.A = \sum Column(B, i).Row(A, i)$

```
> map(i->Column(B,i).Row(A,i),[$1..3]);
  Equal(B.A, add(i, i=%));
```

$$\left[\begin{bmatrix} b_{1,1}\, a_{1,1} & b_{1,1}\, a_{1,2} \\ b_{2,1}\, a_{1,1} & b_{2,1}\, a_{1,2} \\ b_{3,1}\, a_{1,1} & b_{3,1}\, a_{1,2} \\ b_{4,1}\, a_{1,1} & b_{4,1}\, a_{1,2} \end{bmatrix}, \begin{bmatrix} b_{1,2}\, a_{2,1} & b_{1,2}\, a_{2,2} \\ b_{2,2}\, a_{2,1} & b_{2,2}\, a_{2,2} \\ b_{3,2}\, a_{2,1} & b_{3,2}\, a_{2,2} \\ b_{4,2}\, a_{2,1} & b_{4,2}\, a_{2,2} \end{bmatrix}, \begin{bmatrix} b_{1,3}\, a_{3,1} & b_{1,3}\, a_{3,2} \\ b_{2,3}\, a_{3,1} & b_{2,3}\, a_{3,2} \\ b_{3,3}\, a_{3,1} & b_{3,3}\, a_{3,2} \\ b_{4,3}\, a_{3,1} & b_{4,3}\, a_{3,2} \end{bmatrix} \right]$$

$$\mathit{true} \qquad \textbf{(2.3.21)}$$

Elementarmatrizen

MATH: Eine Matrix $B \in K^{m \times m}$ heißt **invertierbar**, falls die induzierte Abbildung

$$\phi_B : K^{m \times 1} \to K^{m \times 1} : S \mapsto B.S$$

bijektiv ist. In diesem Falle wird die inverse Abbildung auch von einer (eindeutig bestimmten) Matrix induziert, welche mit B^{-1} bezeichnet wird. Sie hat die Eigenschaft:

$$B.B^{-1} = B^{-1}.B = I_m := m \times m - \text{Einheitsmatix.}$$

Klar: Die Einheitsmatrix induziert die Identität(sabbildung) von $K^{m \times 1}$:

KOMMENTAR: Die Existenz einer Matrix für die inverse Abbildung ist nicht sofort klar und du wirst den Beweis später in der linearen Algebra sehen.

```
> IdentityMatrix(3);
  Matrix(3)+1;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \textbf{(2.4.1)}$$

Da wir erkannt haben, dass Matrizen Abbildungen induzieren und die Bestimmung der Lösungsmenge einen linearen Gleichungssystems nichts anderes als die Suche nach einer Faser der induzierten Abbildung ist (**DENKANSTOSS**: Warum?), liegt folgende Strategie nahe:

STRATEGIE: Ein lineares Gleichungssystem $A.x = b$ mit $A \in K^{n \times m}$, $b \in K^{n \times 1}$ hat dieselben Lösungen wie

$$B.A.x = B.b$$

für jede invertierbare Matrix $B \in K^{m \times m}$. Die Idee ist, dass beim neuen Gleichungssystem die Dinge einfacher zu sehen sind.

NOTATION: Statt mit A und b getrennt zu operieren, schreiben wir b als weitere Spalte hinter A und haben so eine $(m \times (n+1))$-Matrix, **die erweiterte Matrix des Gleichungssystems**.

Wir benutzen 3 Typen von von invertiertbaren Matrizen. Hier der erste Typ:

```
> AD:=proc(m::posint,r, i::posint,j::posint)
    local II;
    if (i>m) or (j>m) or (i=j) then
     error "Falsche Eingabe"
    end if;
    II := Matrix(m)+1;
    II[i,j] := r;
    return II;
```

```
end proc:
```

```
> AD(3,v,3,2);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & v & 1 \end{bmatrix} \tag{2.4.2}$$

Multipliziert man diese Matrix von links an eine Matrix A heran, so ändert sich die dritte Zeile von A wie folgt, während alle anderen Zeilen unberührt bleiben:

```
> A:=Matrix(3,2,symbol=a);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix} \tag{2.4.3}$$

```
> AD(3,v,3,2).A;
```

$$\begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ v\, a_{2,1} + a_{3,1} & v\, a_{2,2} + a_{3,2} \end{bmatrix} \tag{2.4.4}$$

d. h. die dritte Zeile ist:

(dritte Zeile von A) + v-mal zweite Zeile von A.

DENKANSTOSS: Zeige: Die Matrix *AD(3,v,3,2)* ist invertierbar, indem du die inverse Matrix wieder mit Hilfe des Programmes AD ausdrückst.

Betrachten wir den zweiten Typ von Umformungsmatrizen:

```
> MU:=proc(m::posint,r,i::posint)
    local II;
    if (i>m) or (r=0) then
      error "Falsche Eingabe"
    end if;
    II := Matrix(m,m)+1;
    II[i,i] := r;
    return II;
  end proc:
```

```
> MU(3,r,2);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & r & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{2.4.5}$$

```
> A, MU(3,r,2).A;
```

$$\begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix}, \begin{bmatrix} a_{1,1} & a_{1,2} \\ r\,a_{2,1} & r\,a_{2,2} \\ a_{3,1} & a_{3,2} \end{bmatrix} \qquad \textbf{(2.4.6)}$$

Damit ist klar, welche Zeilenumformungen diese Matrix macht.

DENKANSTOSS: Zeige: Die Matrix *MU(3,r,2)* ist invertierbar, indem du die inverse Matrix wieder mit Hilfe des Programms MU ausdrückst.

Schließlich noch der dritte Typ: die Vertauschung von zwei Zeilen in der Matrix:

```
> VE:=proc(m::posint, i::posint, j::posint)
    local II,k;
    if i>m or j>m or i=j then
      error "Falsche Eingabe";
    end if;
    II:=Matrix(m,m)+1;
    II[i,i] := 0;
    II[j,j] := 0;
    II[i,j] := 1;
    II[j,i] := 1:
    return II;
  end proc:
> VE(3,1,2);VE(3,1,2).VE(3,1,2);
```

$$\begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \textbf{(2.4.7)}$$

MATH: Die quadratischen Matrizen AD, MU, VE unterscheiden sich von einer Einheitsmatrix nur entweder durch die Änderung eines einzigen Eintrags oder durch Vertauschen zweier Zeilen. Solche Matrizen nennt man **elementare Matrizen**.

Mit diesen invertierbaren Matrizen ist es leicht möglich, jedes lineare Gleichungssystem auf sogenannte **strikte Stufengestalt** zu bringen.

MATH: Sei $M \in \mathbb{R}^{m \times n}$ eine Matrix. Für $1 \le i \le m$ ist der i-te **Stufenindex** $St_i(M)$ definiert als:

$$St_i(M) := \min(\{j \in \{1,.., n\} | M_{i,j} \neq 0\}).$$

Falls die Menge auf der rechten Seite leer ist, so definieren wir $St_i(M) := i+n$.
Die Matrix M ist in **Stufenform**, falls die Folge

$$St(M) := (St_1(M), St_2(M), \ldots, St_m(M))$$

streng monoton steigend ist.
Falls zusätzlich noch für jeden Stufenindex $j := St_i(M) \leq n$ für die entsprechende Spalte von M gilt: $M_{-,j} = (I_n)_{-,j}$, so ist M in **strikter Stufenform**.

ÜBUNG [05]:

Schreibe

$$(AD(4, s, 3, 4) \cdot MU(4, t, 2) \cdot VE(4, 1, 3))^{-1}$$

als Produkt von elementaren Matrizen.
(Hinweis: Denke über die Reihenfolge nach, in welcher man morgens die Kleider anzieht und abends auszieht.)

Der Gaußsche Algorithmus

BEISPIEL:Wir begnügen uns vorerst mit einem Beispiel. Wir wollen das lineare Gleichungssystem $Ax=b$ lösen, wobei:

```
> A:=Matrix([[-41,-123,  41, -34, -75, -56,-41],
>            [20,   60, -20, -62, -42,  -8, 20],
>            [-7,  -21,   7, -90, -97, -50, -7],
>            [16,   48, -16, -21,  -5,  30, 16]]);
```

$$A := \begin{bmatrix} -41 & -123 & 41 & -34 & -75 & -56 & -41 \\ 20 & 60 & -20 & -62 & -42 & -8 & 20 \\ -7 & -21 & 7 & -90 & -97 & -50 & -7 \\ 16 & 48 & -16 & -21 & -5 & 30 & 16 \end{bmatrix} \quad (2.5.1)$$

```
> b := Vector(4,[-131, -50, -147, 25]);
```

$$b := \begin{bmatrix} -131 \\ -50 \\ -147 \\ 25 \end{bmatrix} \quad (2.5.2)$$

Wir fügen A und b zu einer Matrix M zusammen und benutzen die oben definierten elementaren Matrizen, um M auf eine strikte Stufenform zu bringen:

```
> M:=<A|b>;
```

(2.5.3)

$$M := \begin{bmatrix} -41 & -123 & 41 & -34 & -75 & -56 & -41 & -131 \\ 20 & 60 & -20 & -62 & -42 & -8 & 20 & -50 \\ -7 & -21 & 7 & -90 & -97 & -50 & -7 & -147 \\ 16 & 48 & -16 & -21 & -5 & 30 & 16 & 25 \end{bmatrix} \tag{2.5.3}$$

(Eintrag normieren)

```
> MU(4,1/(-41),1).M;
```

$$\begin{bmatrix} 1 & 3 & -1 & \frac{34}{41} & \frac{75}{41} & \frac{56}{41} & 1 & \frac{131}{41} \\ 20 & 60 & -20 & -62 & -42 & -8 & 20 & -50 \\ -7 & -21 & 7 & -90 & -97 & -50 & -7 & -147 \\ 16 & 48 & -16 & -21 & -5 & 30 & 16 & 25 \end{bmatrix} \tag{2.5.4}$$

(Vielfache der ersten Zeile von den weiteren Zeilen abziehen)

```
> AD(4,-20,2,1).MU(4,1/(-41),1).M;
```

$$\begin{bmatrix} 1 & 3 & -1 & \frac{34}{41} & \frac{75}{41} & \frac{56}{41} & 1 & \frac{131}{41} \\ 0 & 0 & 0 & -\frac{3222}{41} & -\frac{3222}{41} & -\frac{1448}{41} & 0 & -\frac{4670}{41} \\ -7 & -21 & 7 & -90 & -97 & -50 & -7 & -147 \\ 16 & 48 & -16 & -21 & -5 & 30 & 16 & 25 \end{bmatrix} \tag{2.5.5}$$

```
> AD(4,7,3,1).AD(4,-20,2,1).MU(4,1/(-41),1).M;
```

$$\begin{bmatrix} 1 & 3 & -1 & \frac{34}{41} & \frac{75}{41} & \frac{56}{41} & 1 & \frac{131}{41} \\ 0 & 0 & 0 & -\frac{3222}{41} & -\frac{3222}{41} & -\frac{1448}{41} & 0 & -\frac{4670}{41} \\ 0 & 0 & 0 & -\frac{3452}{41} & -\frac{3452}{41} & -\frac{1658}{41} & 0 & -\frac{5110}{41} \\ 16 & 48 & -16 & -21 & -5 & 30 & 16 & 25 \end{bmatrix} \tag{2.5.6}$$

```
> M1:=AD(4,-16,4,1).AD(4,7,3,1).AD(4,-20,2,1).MU(4,1/(-41),1).
  M;
```

(2.5.7)

$$M1 := \begin{bmatrix} 1 & 3 & -1 & \frac{34}{41} & \frac{75}{41} & \frac{56}{41} & 1 & \frac{131}{41} \\ 0 & 0 & 0 & -\frac{3222}{41} & -\frac{3222}{41} & -\frac{1448}{41} & 0 & -\frac{4670}{41} \\ 0 & 0 & 0 & -\frac{3452}{41} & -\frac{3452}{41} & -\frac{1658}{41} & 0 & -\frac{5110}{41} \\ 0 & 0 & 0 & -\frac{1405}{41} & -\frac{1405}{41} & \frac{334}{41} & 0 & -\frac{1071}{41} \end{bmatrix} \quad (2.5.7)$$

(Normieren)

```
> MU(4,M1[2,4]^(-1),2).M1;
```

$$\begin{bmatrix} 1 & 3 & -1 & \frac{34}{41} & \frac{75}{41} & \frac{56}{41} & 1 & \frac{131}{41} \\ 0 & 0 & 0 & 1 & 1 & \frac{724}{1611} & 0 & \frac{2335}{1611} \\ 0 & 0 & 0 & -\frac{3452}{41} & -\frac{3452}{41} & -\frac{1658}{41} & 0 & -\frac{5110}{41} \\ 0 & 0 & 0 & -\frac{1405}{41} & -\frac{1405}{41} & \frac{334}{41} & 0 & -\frac{1071}{41} \end{bmatrix} \quad (2.5.8)$$

(Abziehen)

```
> AD(4,-M1[1,4],1,2).MU(4,M1[2,4]^(-1),2).M1;
```

$$\begin{bmatrix} 1 & 3 & -1 & 0 & 1 & \frac{1600}{1611} & 1 & \frac{3211}{1611} \\ 0 & 0 & 0 & 1 & 1 & \frac{724}{1611} & 0 & \frac{2335}{1611} \\ 0 & 0 & 0 & -\frac{3452}{41} & -\frac{3452}{41} & -\frac{1658}{41} & 0 & -\frac{5110}{41} \\ 0 & 0 & 0 & -\frac{1405}{41} & -\frac{1405}{41} & \frac{334}{41} & 0 & -\frac{1071}{41} \end{bmatrix} \quad (2.5.9)$$

```
> AD(4,-M1[3,4],3,2).AD(4,-M1[1,4],1,2).MU(4,M1[2,4]^(-1),2).
  M1;
```

(2.5.10)

$$\begin{bmatrix} 1 & 3 & -1 & 0 & 1 & \frac{1600}{1611} & 1 & \frac{3211}{1611} \\ 0 & 0 & 0 & 1 & 1 & \frac{724}{1611} & 0 & \frac{2335}{1611} \\ 0 & 0 & 0 & 0 & 0 & -\frac{4190}{1611} & 0 & -\frac{4190}{1611} \\ 0 & 0 & 0 & -\frac{1405}{41} & -\frac{1405}{41} & \frac{334}{41} & 0 & -\frac{1071}{41} \end{bmatrix} \quad (2.5.10)$$

```
> M2:=AD(4,-M1[4,4],4,2).AD(4,-M1[3,4],3,2).AD(4,-M1[1,4],1,2)
  .MU(4,M1[2,4]^(-1),2).M1;
```

$$M2 := \begin{bmatrix} 1 & 3 & -1 & 0 & 1 & \frac{1600}{1611} & 1 & \frac{3211}{1611} \\ 0 & 0 & 0 & 1 & 1 & \frac{724}{1611} & 0 & \frac{2335}{1611} \\ 0 & 0 & 0 & 0 & 0 & -\frac{4190}{1611} & 0 & -\frac{4190}{1611} \\ 0 & 0 & 0 & 0 & 0 & \frac{37934}{1611} & 0 & \frac{37934}{1611} \end{bmatrix} \quad (2.5.11)$$

(Normieren)

```
> MU(4,M2[3,6]^(-1),3).M2;
```

$$\begin{bmatrix} 1 & 3 & -1 & 0 & 1 & \frac{1600}{1611} & 1 & \frac{3211}{1611} \\ 0 & 0 & 0 & 1 & 1 & \frac{724}{1611} & 0 & \frac{2335}{1611} \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & \frac{37934}{1611} & 0 & \frac{37934}{1611} \end{bmatrix} \quad (2.5.12)$$

(Abziehen)

```
> M3:=AD(4,-M2[1,6],1,3).AD(4,-M2[2,6],2,3).AD(4,-M2[4,6],4,3)
  .MU(4,M2[3,6]^(-1),3).M2;
```

$$M3 := \begin{bmatrix} 1 & 3 & -1 & 0 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad (2.5.13)$$

Jetzt ist die strikte Stufenform erreicht, bei der man die Lösungen ablesen kann. Dies geht so:
Wir entfernen die Nullzeile:

```
> N3:=SubMatrix(M3,[1,2,3],[$1..8]);
```

$$N3 := \begin{bmatrix} 1 & 3 & -1 & 0 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \end{bmatrix} \quad \textbf{(2.5.14)}$$

Für jeden der Nicht-Stufenindizes führen wir eine neue Zeile aus der folgenden Parametermatrix ein, die den fehlenden Parameter benennt:

```
> PP:=Matrix(<Matrix(7,7)+1|Vector(7,i->p[i])>);
```

$$PP := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & p_1 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & p_4 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & p_6 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \quad \textbf{(2.5.15)}$$

```
> N4:=Matrix(<N3,SubMatrix(PP,[2,3,5,7],[$1..8])>);
```

$$N4 := \begin{bmatrix} 1 & 3 & -1 & 0 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \quad \textbf{(2.5.16)}$$

Jetzt sortieren wir die Zeilen:

```
> N5:=VE(7,5,6).VE(7,3,5).VE(7,2,4).N4;
```

$$N5 := \begin{bmatrix} 1 & 3 & -1 & 0 & 1 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \quad \textbf{(2.5.17)}$$

Nun räumen wir den Rest aus, bei der vorletzten Spalte angefangen:

```
> AD(7,-1,1,7).N5;
```

$$\begin{bmatrix} 1 & 3 & -1 & 0 & 1 & 0 & 0 & 1-p_7 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \tag{2.5.18}$$

```
> AD(7,-1,4,5).AD(7,-1,1,5).AD(7,-1,1,7).N5;
```

$$\begin{bmatrix} 1 & 3 & -1 & 0 & 0 & 0 & 0 & 1-p_5-p_7 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1-p_5 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \tag{2.5.19}$$

```
> AD(7,1,1,3).AD(7,-1,4,5).AD(7,-1,1,5).AD(7,-1,1,7).N5;
```

$$\begin{bmatrix} 1 & 3 & 0 & 0 & 0 & 0 & 0 & 1+p_3-p_5-p_7 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1-p_5 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \tag{2.5.20}$$

```
> N6:=AD(7,-3,1,2).AD(7,1,1,3).AD(7,-1,4,5).AD(7,-1,1,5).AD(7,
  -1,1,7).N5;
```

(2.5.21)

$$N6 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 1-3\,p_2+p_3-p_5-p_7 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & p_2 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & p_3 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1-p_5 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & p_5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & p_7 \end{bmatrix} \tag{2.5.21}$$

Die letzte Spalte sollte die allgemeine Lösung des ursprünglichen Gleichungssystems sein. Wir prüfen wenigstens, ob es wirklich Lösungen sind, die wir bestimmt haben:

```
> A.Column(N6,8)-b;
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \tag{2.5.22}$$

ÜBUNG [06]:

Sei **n** deine Matrikelnummer.

```
Aufgabe6Mat:=proc(n::posint)
  local T1a,T1b,T2a,T2b,M;
   T2a:=Matrix(4,4,(i,j)->if i>j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^2+j^3-j+5) fi);
   T2b:=Matrix(4,4,(i,j)->if i<j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^3+j^2-j+5) fi);
   T1a:=Matrix(3,3,(i,j)->if i<j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^2+j^3-j-i+6) fi);
   T1b:=Matrix(3,3,(i,j)->if i>j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^3+j^2-i+5) fi);
  M:=T1b.T1a.<Matrix(3,3)+1|Matrix(3,1)>.T2a.T2b;
  convert(convert(M,listlist) mod iquo(n,1000),Matrix);
end proc:
```

Führe den Gaußschen Algorithmus für **Aufgabe6Mat(n)** (über $\mathbb{Q}$) und unten angegebenes **b** analog zum obigen Beispiel durch.

```
n:=;
Error, `:` unexpected
> Aufgabe6Mat(n);
Error, invalid input: Aufgabe6Mat expects its 1st argument, n, to be of type posint, but received n
> b:=<42,42,42>;
```

$$b := \begin{bmatrix} 42 \\ 42 \\ 42 \end{bmatrix} \quad (2.5.23)$$

Wir wollen den Gaußschen Algorithmus nutzen, um die Inverse einer invertierbaren Matrix zu bestimmen. Also jener Matrix, welche die Umkehrabbildung der induzierten Abbildung induziert:

```
> B := Matrix( [ [3,-5,4] , [-4,5,-2] , [1,2,-5] ] );
```

$$B := \begin{bmatrix} 3 & -5 & 4 \\ -4 & 5 & -2 \\ 1 & 2 & -5 \end{bmatrix} \quad (2.5.24)$$

Dafür nutzen wir folgenden Trick:

Sei $C := \begin{bmatrix} c_{1,1} & c_{1,2} & c_{1,3} \\ c_{2,1} & c_{2,2} & c_{2,3} \\ c_{3,1} & c_{3,2} & c_{3,3} \end{bmatrix}$ die inverse Matrix von B. Aus $B.C = I_3$ und der Tatsache, dass die j-te Spalte von $B \cdot C$ gerade B mal j-te Spalte von C ist, folgt, dass wir die 3 linearen Gleichungssysteme

$$B.x_1 = \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \; B.x_2 = \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \; B.x_3 = \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

lösen müssen und damit die Spalten x_i von C bestimmen können.

Da die Umformungen nur durch die Matrix B bestimmt sind, können wir mit allen 3 rechten Seiten gleichzeitig arbeiten und erhalten die erweiterte Matrix:

```
> G:=<B|Matrix(3,3)+1>;
```

$$G := \begin{bmatrix} 3 & -5 & 4 & 1 & 0 & 0 \\ -4 & 5 & -2 & 0 & 1 & 0 \\ 1 & 2 & -5 & 0 & 0 & 1 \end{bmatrix} \quad (2.5.25)$$

(Vertauschen)

```
> VE(3,1,3).G;
```

$$\begin{bmatrix} 1 & 2 & -5 & 0 & 0 & 1 \\ -4 & 5 & -2 & 0 & 1 & 0 \\ 3 & -5 & 4 & 1 & 0 & 0 \end{bmatrix} \quad (2.5.26)$$

(Abziehen)

```
> AD(3,-3,3,1).AD(3,4,2,1).VE(3,1,3).G;
```

$$\begin{bmatrix} 1 & 2 & -5 & 0 & 0 & 1 \\ 0 & 13 & -22 & 0 & 1 & 4 \\ 0 & -11 & 19 & 1 & 0 & -3 \end{bmatrix} \tag{2.5.27}$$

(Normieren)

```
> MU(3,1/13,2).AD(3,-3,3,1).AD(3,4,2,1).VE(3,1,3).G;
```

$$\begin{bmatrix} 1 & 2 & -5 & 0 & 0 & 1 \\ 0 & 1 & -\frac{22}{13} & 0 & \frac{1}{13} & \frac{4}{13} \\ 0 & -11 & 19 & 1 & 0 & -3 \end{bmatrix} \tag{2.5.28}$$

(Abziehen)

```
> AD(3,11,3,2).AD(3,-2,1,2).MU(3,1/13,2).AD(3,-3,3,1).AD(3,4,
  2,1).VE(3,1,3).G;
```

$$\begin{bmatrix} 1 & 0 & -\frac{21}{13} & 0 & -\frac{2}{13} & \frac{5}{13} \\ 0 & 1 & -\frac{22}{13} & 0 & \frac{1}{13} & \frac{4}{13} \\ 0 & 0 & \frac{5}{13} & 1 & \frac{11}{13} & \frac{5}{13} \end{bmatrix} \tag{2.5.29}$$

(Normieren)

```
> MU(3,13/5,3).AD(3,11,3,2).AD(3,-2,1,2).MU(3,1/13,2).AD(3,-3,
  3,1).AD(3,4,2,1).VE(3,1,3).G;
```

$$\begin{bmatrix} 1 & 0 & -\frac{21}{13} & 0 & -\frac{2}{13} & \frac{5}{13} \\ 0 & 1 & -\frac{22}{13} & 0 & \frac{1}{13} & \frac{4}{13} \\ 0 & 0 & 1 & \frac{13}{5} & \frac{11}{5} & 1 \end{bmatrix} \tag{2.5.30}$$

(Abziehen)

```
> AD(3,22/13,2,3).AD(3,21/13,1,3).MU(3,13/5,3).AD(3,11,3,2).AD
  (3,-2,1,2).MU(3,1/13,2).AD(3,-3,3,1).AD(3,4,2,1).VE(3,1,3).
  G;
```

(2.5.31)

$$\begin{bmatrix} 1 & 0 & 0 & \frac{21}{5} & \frac{17}{5} & 2 \\ 0 & 1 & 0 & \frac{22}{5} & \frac{19}{5} & 2 \\ 0 & 0 & 1 & \frac{13}{5} & \frac{11}{5} & 1 \end{bmatrix} \tag{2.5.31}$$

Beachte, dass wir nun **keine** Parameter einführen müssen, da wir mit 3 rechten Seiten gearbeitet haben und dies somit bereits die strikte Stufenform ist.

Wir können die inverse Matrix nun aus der rechten Hälfte ablesen:

```
> C:=Matrix([ [21/5,17/5,2] , [22/5,19/5,2] , [13/5,11/5,1] ])
  ;
```

$$C := \begin{bmatrix} \frac{21}{5} & \frac{17}{5} & 2 \\ \frac{22}{5} & \frac{19}{5} & 2 \\ \frac{13}{5} & \frac{11}{5} & 1 \end{bmatrix} \tag{2.5.32}$$

```
> B.C;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{2.5.33}$$

```
> C.B;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{2.5.34}$$

ÜBUNG [07]:

Sei **n** deine Matrikelnummer.

```
Aufgabe7Mat:=proc(n::posint)
  local T1a,T1b,T2a,T2b,M;
  T2a:=Matrix(3,3,(i,j)->if i>j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^2+j^3-j+5) fi);
  T2b:=Matrix(3,3,(i,j)->if i<j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^3+j^2-j+5) fi);
  T1a:=Matrix(3,3,(i,j)->if i<j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^2+j^3-j-i+6) fi);
  T1b:=Matrix(3,3,(i,j)->if i>j then 0; elif i=j then 1
else i-j+1+(n*(i-j+1) mod i^3+j^2-i+5) fi);
```

```
    M:=T1b.T1a.T2a.T2b;
    convert(convert(M,listlist) mod iquo(n,1000),Matrix);
  end proc:
```

Führe den Gaußschen Algorithmus für **Aufgabe7Mat(n)** (über $\mathbb{Q}$) zur Bestimmung der inversen Matrix analog zum obigen Beispiel durch.